Pensieve header: Implementing Θ - the main notebook accompanying Talks/Toronto-241030.

pdf

Invariance under R3

pdf

This is Theta.nb of http://drorbn.net/to24/ap.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Talks\\Toronto-241030"];
```

pdf

In[∘]:=
```
Once[<< KnotTheory`; << Rot.m; << PolyPlot.m];
```

pdf

In[∘]:=
```
T3 = T1 T2;
```

pdf

In[∘]:=
```
CF[ℰ_] := Module[{vs = Union@Cases[ℰ, g__, ∞], ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) ⧴ Factor[c] (Times @@ vs^ps)] ];
```

In[∘]:=

$$R_1[s\_, i\_, j\_] = \frac{s}{2} + s\,T_2^s\,g_{1ii}\,g_{2ji} + \frac{s\,(T_1^s-1)\,T_2^{2s}\,g_{1ji}\,g_{2ji}}{T_2^s-1} - s\,g_{1ii}\,g_{2jj} - \frac{s\,(T_1^s-1)\,T_2^s\,g_{1ji}\,g_{2jj}}{T_2^s-1} - s\,g_{3ii} - s\left(T_2^s-1\right)g_{2ji}\,g_{3ii} + 2\,s\,g_{2jj}\,g_{3ii} + \frac{s\,(T_3^s-1)\,g_{3ji}}{T_2^s-1} - \frac{s\,T_2^s\,(T_3^s-1)\,g_{1ii}\,g_{3ji}}{T_2^s-1} - \frac{s\,(T_1^s-1)\,(1+T_2^s)\,(T_3^s-1)\,g_{1ji}\,g_{3ji}}{T_2^s-1} + \frac{s\,(T_3^s-1)\,g_{2ij}\,g_{3ji}}{T_2^s-1} + s\left(T_3^s-1\right)g_{2ji}\,g_{3ji} + \frac{s\,(T_2^s-2)\,(T_3^s-1)\,g_{2jj}\,g_{3ji}}{T_2^s-1} + s\,g_{1ii}\,g_{3jj} + \frac{s\,(T_1^s-1)\,T_2^s\,g_{1ji}\,g_{3jj}}{T_2^s-1} - s\,g_{2ii}\,g_{3jj} - s\,T_2^s\,g_{2ji}\,g_{3jj};$$

In[∘]:=

$$\theta[\{s0\_, i0\_, j0\_\}, \{s1\_, i1\_, j1\_\}] := \frac{s1\left(T_1^{s0}-1\right)T_2^{s0}\left(T_3^{s1}-1\right)g_{1j1i0}\,g_{2i1i0}\,g_{3j0i1}}{T_2^{s1}-1} - \frac{s1\left(T_1^{s0}-1\right)\left(T_3^{s1}-1\right)g_{1j1i0}\,g_{2i1j0}\,g_{3j0i1}}{T_2^{s1}-1} - \frac{s1\left(T_1^{s0}-1\right)T_2^{s0}\left(T_3^{s1}-1\right)g_{1j1i0}\,g_{2j1i0}\,g_{3j0i1}}{T_2^{s1}-1} + \frac{s1\left(T_1^{s0}-1\right)\left(T_3^{s1}-1\right)g_{1j1i0}\,g_{2j1j0}\,g_{3j0i1}}{T_2^{s1}-1}$$

pdf

In[∘]:=

$$R_{11}[\{s\_, i\_, j\_\}] = CF\Big[s\,\Big(1/2 - g_{3ii} + T_2^s\,g_{1ii}\,g_{2ji} - g_{1ii}\,g_{2jj} - \left(T_2^s-1\right)g_{2ji}\,g_{3ii} + 2\,g_{2jj}\,g_{3ii} - \left(1-T_3^s\right)g_{2ji}\,g_{3ji} - g_{2ii}\,g_{3jj} - T_2^s\,g_{2ji}\,g_{3jj} + g_{1ii}\,g_{3jj} + \Big(\left(T_1^s-1\right)g_{1ji}\left(T_2^{2s}\,g_{2ji} - T_2^s\,g_{2jj} + T_2^s\,g_{3jj}\right) + \left(T_3^s-1\right)g_{3ji}\left(1 - T_2^s\,g_{1ii} - \left(T_1^s-1\right)\left(T_2^s+1\right)g_{1ji} + \left(T_2^s-2\right)g_{2jj} + g_{2ij}\right)\Big) \Big/ \left(T_2^s-1\right)\Big)\Big];$$

pdf

In[∘]:=

$$R_{12}[\{s0\_, i0\_, j0\_\}, \{s1\_, i1\_, j1\_\}] := CF\Big[s1\left(T_1^{s0}-1\right)\left(T_2^{s1}-1\right)^{-1}\left(T_3^{s1}-1\right)g_{1,j1,i0}\,g_{3,j0,i1}\left(\left(T_2^{s0}\,g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0}\,g_{2,j1,i0} - g_{2,j1,j0}\right)\right)\Big]$$

pdf

In[∘]:=
```
Γ₁[φ_, k_] = -φ / 2 + φ g_{3kk};
```

pdf

In[∘]:=
```
δ_{i_,j_} := If[i === j, 1, 0];
gR_{s_,i_,j_} := {
  g_{ν_jβ_} :→ g_{νj⁺β} + δ_{jβ}, g_{ν_iβ_} :→ T_ν^s g_{νi⁺β} + (1 - T_ν^s) g_{νj⁺β} + δ_{iβ},
  g_{ν_α_i⁺} :→ T_ν^s g_{ναi} + δ_{αi⁺}, g_{ν_α_j⁺} :→ g_{ναj} + (1 - T_ν^s) g_{ναi} + δ_{αj⁺}
 }
```

Proof of Reidemeister 3:

pdf

In[∘]:=
```
DSum[Cs___] := Sum[R₁₁[c], {c, {Cs}}] + Sum[R₁₂[c0, c1], {c0, {Cs}}, {c1, {Cs}}]
lhs = DSum[{1, j, k}, {1, i, k⁺}, {1, i⁺, j⁺}, {s, m, n}] //. gR_{1,j,k} ⋃ gR_{1,i,k⁺} ⋃ gR_{1,i⁺,j⁺};
rhs = DSum[{1, i, j}, {1, i⁺, k}, {1, j⁺, k⁺}, {s, m, n}] //. gR_{1,i,j} ⋃ gR_{1,i⁺,k} ⋃ gR_{1,j⁺,k⁺};
Simplify[lhs == rhs]
```

Out[∘]=
pdf

True

tex

\needspace{30mm}

pdf

The Main Program

pdf

In[∘]:=
```
Θ[K_] := Module[{Cs, φ, n, A, Δ, G, ev, θ},
  {Cs, φ} = Rot[K]; n = Length[Cs];
  A = IdentityMatrix[2 n + 1];
  Cases[Cs, {s_, i_, j_} :→ (A⟦{i, j}, {i + 1, j + 1}⟧ += ( -T^s  T^s - 1
                                                              0     -1    ))];
  Δ = T^((-Total[φ]-Total[Cs⟦All,1⟧])/2) Det[A];
  G = Inverse[A];
  ev[ℰ_] := Factor[ℰ /. g_{ν_,α_,β_} :→ (G⟦α, β⟧ /. T → T_ν)];
  θ = ev[∑_{k1=1}^n ∑_{k2=1}^n R₁₂[Cs⟦k1⟧, Cs⟦k2⟧]];
  θ += ev[∑_{k=1}^n R₁₁[Cs⟦k⟧]];
  θ += ev[∑_{k=1}^{2n} Γ₁[φ⟦k⟧, k]];
  Factor@{Δ, (Δ /. T → T₁) (Δ /. T → T₂) (Δ /. T → T₃) θ}];
```

The Rolfsen Table

In[∘]:=
```
tab250 = {{1, 0}}~Join~Table[Θ[K], {K, AllKnots[{3, 10}]}];
```

KnotTheory: Loading precomputed data in PD4Knots`.

In[ ]:=
```
g250 = GraphicsGrid[Partition[PolyPlot /@ tab250, 25], Spacings → 0, Dividers → All]
```
Out[ ]=



In[ ]:=
```
Export["g250.png", g250]
```
Out[ ]=
```
g250.png
```

pdf

The Trefoil, Conway, and Kinoshita-Terasaka

tex

```
\parpic[r]{\parbox{15mm}{
 \includegraphics[width=15mm]{K11n34.png}
 \vskip 1mm
 \includegraphics[width=15mm]{K11n42.png}
}}
```



pdf

In[∘]:= `Θ[Knot[3, 1]] // Expand`

Out[∘]=
pdf

$$\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2 T_2^2}+\frac{1}{T_1 T_2^2}+\frac{1}{T_1^2 T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2 T_2-T_2^2+T_1 T_2^2-T_1^2 T_2^2\right\}$$

pdf

In[∘]:= `GraphicsRow[PolyPlot[Θ[Knot[#]]] & /@ {"3_1", "K11n34", "K11n42"}]`

Out[∘]=
pdf



tex

(Note that the genus of the Conway knot appears to be bigger than the genus of Kinoshita-Terasaka) pdf

Some Torus Knots

pdf

In[∘]:=
```
TKs = {{13, 2}, {17, 3}, {13, 5}, {7, 6}};
GraphicsRow[PolyPlot[Θ[TorusKnot @@ #]] & /@ TKs]
GraphicsRow[TubePlot[TorusKnot @@ #] & /@ TKs]
```

Out[∘]=
pdf



Out[∘]=
pdf



The 132-crossing Torus Knot $T_{22/7}$

In[∘]:=
```
AbsoluteTiming[T227 = Θ[TorusKnot[22, 7]];]
```

Out[∘]=

{698.928, Null}

In[*]:=
```
T227Plot = ImageCompose[
  PolyPlot[T227],
  TubePlot[TorusKnot[22, 7], ImageSize → 240],
  {Right, Bottom}, {Right, Bottom}
 ]
```
Out[*]=



In[*]:=
```
Export["T227Plot.pdf", T227Plot]
```
Out[*]=

T227Plot.pdf

Large Random Knots

See also https://drorbn.net/AcademicPensieve/Projects/HigherRank/DunfieldKnots/.

In[∘]:=
```
DK300 = PolyPlot@Expand[
   Get["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank\\DunfieldKnots\\D300.m"][[
     2]] /. {T1 → T1, T2 → T2}]
```

Out[∘]=



In[∘]:=
```
Export["DK300.png", DK300, ImagePadding → None, PlotRangePadding → None]
```

Out[∘]=

DK300.png

In[∘]:=
```
Export["DK300.pdf", DK300]
```

Out[∘]=

DK300.pdf

Some Virtual Knots

In[ ]:=
```
pd = PD[X[2, 4, 3, 1], X[3, 1, 4, 2]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[ ]=

{{{-1, 4, 2}, {-1, 1, 3}}, {0, 0, 1, 0}}

Out[ ]=

$$\left\{\frac{1}{\sqrt{T}}, \frac{-1-2 T_2+4 T_1 T_2}{2 T_1 T_2}\right\}$$

Out[ ]=



In[∘]:=
```
pd = PD[X[3, 8, 4, 1], X[1, 6, 2, 7], X[5, 2, 6, 3], X[7, 5, 8, 4]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[∘]=

{{{-1, 8, 3}, {-1, 6, 1}, {-1, 2, 5}, {1, 4, 7}}, {0, 0, 0, 0, 1, -1, -1, -1}}

Out[∘]=

$$\left\{-\frac{1-3T+T^2}{T},\ -\frac{\left(1-3T_1+T_1^2\right)\left(-1+T_1+2T_1^2+T_2-7T_1^2T_2-2T_1^3T_2-T_1T_2^2+4T_1^2T_2^2+5T_1^3T_2^2-3T_1^2T_2^3+T_1^3T_2^3\right)}{T_1^2}\right\}$$

Out[∘]=



In[ ]:=
```
pd = PD[X[2, 8, 3, 1], X[3, 1, 4, 2], X[6, 4, 7, 5], X[7, 5, 8, 6]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[ ]=
```
{{{-1, 8, 2}, {-1, 1, 3}, {-1, 4, 6}, {-1, 5, 7}}, {0, 0, 1, 0, 0, 1, 1, 0}}
```

Out[ ]=

$$\left\{\frac{1}{\sqrt{T}}, \frac{1}{2\,T_1^3\,T_2^4}\left(4 - 4\,T_1 - 6\,T_2 + 2\,T_1\,T_2 + 6\,T_1^2\,T_2 + 4\,T_1\,T_2^2 - 8\,T_1^2\,T_2^2 + 2\,T_2^3 + 2\,T_1\,T_2^3 + T_1^2\,T_2^3 - 4\,T_1^3\,T_2^3 - 2\,T_1\,T_2^4 - 6\,T_1^2\,T_2^4 + 10\,T_1^3\,T_2^4\right)\right\}$$

Out[ ]=